
Writing templates can be tricky, especially when it comes to 
creating #AT statements that correspond to ABC virtual method 
embeds. Here's a template that makes that task easy (available 
for download at the end of the article). I based this code on 
something Lee White wrote for Clarion4 (used with permission - 
thanks, Lee!). 

The template is in a TPW, so you'll need to add the statement 
#INCLUDE('EMBEDS.TPW') to an appropriate TPL file. Once you've 
done this, open any application, go to embed view, locate the 
embed you want to generate an #AT statement for, and choose 
the Embed_Info code template, as shown in Figure 1. 

Figure 1. Selecting the Embed_Info code template.

Figure 2. The Embed_Info code template in the embed 
point.

Once you've inserted the template, go to Source View (the easiest 
way is to click the Source button on the Embeds window). In the 
example shown in Figures 1 and 2, the template generates the 
following code (line break added):

!>> #AT(%BrowserMethodCodeSection,'1','ApplyFilter',
          '()'),PRIORITY( ???? )
!>> !! What follows is debugging info
!>> EmbedID........... = %BrowserMethodCodeSection
!>> EmbedParameters... = 1, ApplyFilter, ()
!>> Description....... = Browser Method Code Section
!>> Process........... = ProcessClass
!>> Object............ = BRW1
!>> TwoParams ........ = 2
!>> ThreeParams....... = 15

Just copy and paste the generated #AT statement for use in your 
own templates. The template source looks like this (line breaks 
added): 

#!
#!TEMPLATE(ABC_EmbedDefs,'Embed definitions')
    ,FAMILY('ABC')
#!--------------------------------------------
#CODE(Embed_Info,'Embed definition and info')
#!--------------------------------------------
#DECLARE(%TwoParams,LONG)

#DECLARE(%ThirdParams,LONG)
#DECLARE(%ParamString,STRING)
#IF(%EmbedParameters)
#CLEAR(%TwoParams)
#SET(%TwoParams,INSTRING(',',%EmbedParameters,1,1))
#IF(%TwoParams)
#IF(SUB(%EmbedParameters,(%TwoParams+2),1) <> '(') 
#SET(%ThirdParams,INSTRING(',',
    %EmbedParameters,1,%TwoParams+2))
#SET(%ParamString,'''' & SUB(%EmbedParameters,1,
    (%TwoParams-1)) & ''',''' & SUB(%EmbedParameters,
    (%TwoParams+2),((%ThirdParams) - (%TwoParams+2))) 
    & ''',''' & SUB(%EmbedParameters,(%ThirdParams+2)
    ,LEN(%EmbedParameters)) & '''')
#ELSE
#SET(%ParamString,'''' & SUB(%EmbedParameters,1,
    (%TwoParams-1)) & ''',''' & SUB(%EmbedParameters,
    (%TwoParams+2),LEN(%EmbedParameters)) & '''')
#END
!>> #AT(%EmbedID,%ParamString),PRIORITY( ???? )
#ELSE
!>> #AT(%EmbedID,'%EmbedParameters'),PRIORITY( ???? )
#ENDIF
#ELSE
!>> #AT(%EmbedID),PRIORITY( ???? )
#ENDIF
!>> !! What follows is debugging info
!>> EmbedID........... = %EmbedID
!>> EmbedParameters... = %EmbedParameters
!>> Description....... = %EmbedDescription
!>> Process........... = %ProcessType
!>> Object............ = %ThisObjectName
!>> TwoParams ........ = %TwoParams
!>> ThreeParams....... = %ThirdParams
 

You can also download the template here.

Andrew Guidroz II, when he isn't handfeeding the tufted titmouse, writes software 
for all facets of the insurance industry. His famous Cajun cookouts have become a 
central feature of Clarion conferences throughout the U.S. Andrew's Cajun website 
is www.coonass.com.

